LT1013/LT1014 Macromodel

```
* This more complete macromodel has been adapted from the Parts
* generated LT1013/LT1014 model.  This version features closer
* fidelity to the real part, with input common-mode clamping, and
* compensated output clamping.  It can be used for large signal
* and/or single supply applications, where the inputs can
* potentially be overdriven.  Since it uses more active devices,
* it may run more slowly than will a conventional macromodel.
*
* connections:   non-inverting input
*                | inverting input
*                | | positive power supply
*                | | | negative power supply
*                | | | | output
*                | | | | |
.subckt LT1013   1 2 3 4 5
*
*
  c1   11 12 8.661E-12
  c2    6  7 30.00E-12
  dc    8 53 dx
  de   54  8 dx
  dlp  90 91 dx
  dln  92 90 dx
  dp    4  3 dx
  egnd 99  0 poly(2) (3,0) (4,0) 0 .5 .5
  fb    7 99 poly(5) vb vc ve vlp vln 0 2.475E9 -2E9 2E9 2E9 -2E9
  ga    6  0 11 12 113.1E-6
  gcm   0  6 10 99 225.7E-12
  iee   3 10 dc 12.03E-6
  hlim 90  0 vlim 1K
  q1   11 102 13 qx
  q2   12 101 14 qx
  rb1   2 102 400
  rb2   1 101 400
  dcm1 105 102 dx
  dcm2 105 101 dx
  vcmc 105 4 dc 0.4
  r2    6  9 100.0E3
  rc1   4 11 8.841E3
  rc2   4 12 8.841E3
  re1  13 10 4.519E3
  re2  14 10 4.519E3
  ree  10 99 16.63E6
  ro1   8  5 80
  ro2   7 99 25
  ip    3  4 328E-6
  vb    9  0 dc 0
  vc    3 53 dc 1.610
  ve   54  4 dc .61
  vlim  7  8 dc 0
  vlp  91  0 dc 25
  vln   0 92 dc 25
.model dx D(Is=800.0E-18)
.model qx PNP(Is=800.0E-18 Bf=400)
.ends
```

```
*-------------------------------------
* connections:   non-inverting input
*                | inverting input
*                | | positive power supply
*                | | | negative power supply
*                | | | | output
*                | | | | |
.subckt LT1014   1 2 3 4 5
*
  x_LT1014  1 2 3 4 5 LT1013
.ends
*-------------------------------------
* connections:   non-inverting input
*                | inverting input
*                | | positive power supply
*                | | | negative power supply
*                | | | | output
*                | | | | |
.subckt LT1006   1 2 3 4 5
*
  x_LT1006  1 2 3 4 5 LT1013
.ends
*-------------------------------------
```